Corrigé du TD 3 : Arithmétique et polynômes

Jean Sébastien ROY, 1999.

Ex 1 : Calcul de la n-ième décimale de ln(9/10)

```
> restart:
```

Si : r=10^n mod k

Alors :

```
> (10^n-r)/k=10^n/k-r/k;
```

$$\frac{10^n-r}{k}=\frac{10^n}{k}-\frac{r}{k}$$

est un entier relatif. Et donc: 10^n/k et r/k ont les mêmes parties fractionaires.

Ainsi le calcul de r/k donne les décimales de 1/k à partir de la n-ième.

Exemple :

```
> evalf(1/17,50);
```

$$.058823529411764705882352941176470588235294117647059$$

```
> r:=10^11 mod 17;
```

$$r := 3$$

```
> evalf(r/17,50);
```

$$.17647058823529411764705882352941176470588235294118$$

Mais il vaut mieux écrire :

```
> r:=10 &^ 11 mod 17;
```

$$r := 3$$

Car sous la forme &^, Maple ne calcule pas 10^9 puis 1000 mod 17 mais effectue directement le le calcul entier en ne conservant que des nombres plus petits que 17^2 :
Comme 11=(2*2+1)*2+1 on calcule :
(10^2^2*10)^2*10 mod 17 soit :
10^2 mod 17 = 15
15^2 mod 17 = 4
4*10 mod 17 = 6
6^2 mod 17 = 2
2*10 mod 17 = 3
donc :
10^11 mod 17 = 3
Pour calculer a^b mod c il faut environ ln(b)/ln(2) opérations.

Ainsi, on a rapidement la 100 001-ieme décimale de 1/17 sans avoir à effectuer des calculs sur des nombres de plus de 3 chiffres (au lieu de 100 000).

```
> r:=10 &^ 100000 mod 17;
```

$$r := 1$$

```
> evalf(r/17,10);
```

$$.05882352941$$

```
> evalf(frac(10^100000/17));
```

$$.05882352941$$

Cela marche pareil en base b (il suffit de remplacer 10 par b)

Exemple de calcul : calcul de la n-ième décimale de ln(9/10) :

```
> sum(1/(k*10^k),k=1..infinity);
```

$$-\ln\left(\frac{9}{10}\right)$$

On cherche la partie fractionaire de:

```
> 10^n*Sum(1/(k*10^k),k=1..infinity);
```

$$10^n\left(\sum_{k=1}^{\infty}\frac{1}{k\,10^k}\right)$$

Soit la partie fractionaire de :

```
> Sum(10^(n-k)/k,k=1..n-1)+Sum(10^(n-k)/k,k=n..infinity);
```

$$\left(\sum_{k=1}^{n-1}\frac{10^{(n-k)}}{k}\right)+\left(\sum_{k=n}^{\infty}\frac{10^{(n-k)}}{k}\right)$$

Pour la deuxième somme, pas de problème, il suffit de calculer ses premières décimales tel quel à l'aide de quelques termes :

Pour la première somme, on utilise la méthode décrite plus haut.

On écrit une fonction qui calcule b^n mod j en utilisant la méthode rapide si n>0

```
> deci := proc(b,n,j) if n<=0 then evalf(b^n/j) else evalf((b &^ n mod j)/j) fi:end;
```

$deci$:= **proc**(b, n, j) **if** $n\leq 0$ **then** evalf(b^n/j) **else** evalf((b `&^` n) mod j/j) **fi end**

(Exemple avec n=99)

```
> frac(add(deci(10,99-j,j),j=1..99+10));
```


$$.76602772$$

Attention ! Il faut utiliser **add** et non pas **sum** sinon Maple mémorise tous les nombres sommés et effectue la somme ensuite.

```
> evalf(frac(-10^99*ln(9/10)),110);
```

$$.76602772591$$

Même exemple, mais avec n=9 999 :

```
> frac(add(deci(10,9999-j,j),j=1..9999+10));
```

$$.292780$$

```
> evalf(frac(-10^9999*ln(9/10)),10010);
```

$$.29277671955$$

Calculons quelques décimales de Pi à l'aide de la formule :

```
> 16^N * Pi=Sum(16^(N-k)*(4/(8*k+1)-2/(8*k+4)-1/(8*k+5)-1/(8*k+6)),k=0..inf);
```

$$16^N\,\pi=\sum_{k=0}^{inf}16^{(N-k)}\left(4\,\frac{1}{8\,k+1}-2\,\frac{1}{8\,k+4}-\frac{1}{8\,k+5}-\frac{1}{8\,k+6}\right)$$

Vérifions tout de même :

```
> evalf(sum(16^(-k)*(4/(8*k+1)-2/(8*k+4)-1/(8*k+5)-1/(8*k+6)),k=0..infinity));
```

$$3.141592652$$

```
> n:=1000;
```

$$n := 1000$$

```
> add(4*deci(16,n-j,8*j+1)-2*
  deci(16,n-j,8*j+4)-1*deci(16,n-j,8*j+5)-1*deci(16,n-j,8*j+6),j=0..n+10);
```

$$-58.71115510$$

Ce qui nous intéresse c'est la partie fractionaire. Comme on n'a pas calculé correctement la partie entière (volontairement), le signe de cette partie entière n'est pas bon ce qui modifie la valeur de la partie fractionaire (complément à 10). On rajoute 1 à la partie fractionaire pour avoir la partie fractionaire correcte.

```
> 1+frac(%);
```

$$.28884490$$

Soit affiché en hexadécimal (c'est une notation pratique pour la base 16 ou l'on note 0=0, ..., 9=9, 10=A ... 15=F

```
> convert(trunc(16^8*%),hex);
```

$$\mathit{49F1BD47}$$

Vérification :

```
> t:=frac(evalf(Pi*16^1000,1300)):
> convert(trunc(t*16^8),hex);
```

$$\mathit{49F1C09B}$$

```
>
```

Ex 2 : polynômes interpolateurs

```
> restart:
```

Définissons la fonction :

```
> f:=x -> 1/(x^2+1);
```

$$f := x\rightarrow\frac{1}{x^2+1}$$

Essayons d'interpoler cette fonction avec le développement de Taylor :

```
> taylor(f(x),x=0,5);
```

$$1-x^2+x^4+O(x^6)$$

```
> convert(%,polynom);
```

$$1-x^2+x^4$$

```
> plot([f(x),convert(taylor(f(x),x=0,3),polynom),
  convert(taylor(f(x),x=0,5),polynom)],x=-5..5,-1..1,color=[black,red,blue,green],num
  points=200,thickness=3);
```



Ca diverge très rapidement.
Essayons avec les polynômes interpolateurs de Lagrange. Première définition des xi :

```
> x1:=(i,n) -> -5+10*i/n;
```

$$x1 := (i, n) \to -5+10\,\frac{i}{n}$$

La fonction suivante renvoie le i-ième polynome interpolateur de Lagrange de degré n, pour les points définis par x(i,n) :
Cette fonction ne marche que si x est définie et si i, n ont chacun une valeur.

```
> La:=(i,n,x) -> product((X-x(j,n))/(x(i,n)-x(j,n)),j=0..i-1)*
  product((X-x(j,n))/(x(i,n)-x(j,n)),j=i+1..n);
```

$$La := (i, n, x) \to \left(\prod_{j=0}^{i-1} \frac{X-x(j, n)}{x(i, n)-x(j, n)}\right)\left(\prod_{j=i+1}^{n} \frac{X-x(j, n)}{x(i, n)-x(j, n)}\right)$$

A partir des poynômes La, on construit le polynôme interpolateur en les x(i,n) pour la fonction f :
Les '' autour de La(i,n,x) indiquent à Maple de ne pas calculer La(i,n,x) avant de donner une valeur à i, n et x.

```
> fLa:=(f,n,x) -> sum(f(x(i,n))*'La(i,n,x)',i=0..n);
```

$$fLa := (f, n, x) \to \sum_{i=0}^{n} f(x(i, n))\ \text{'La}(i, n, x)\text{'}$$

Exemple :

```
> fLa(f,2,x1);
```

$$-\frac{1}{130} X \left(-\frac{1}{10} X+\frac{1}{2}\right)+\left(\frac{1}{5} X+1\right)\left(-\frac{1}{5} X+1\right)+\frac{1}{130}\left(\frac{1}{10} X+\frac{1}{2}\right) X$$

```
> plot([f(X),fLa(f,5,x1),fLa(f,15,x1),fLa(f,20,x1)],X=-5..5,-1..1,color=[black,red,bl
  ue,green],numpoints=200,thickness=3);
```



On voit apparaitre un phénomène (appelé phénomène de Runge), au bord de l'intervalle :
Le polynôme interpolateur oscille fortement au bord de l'intervalle, d'autant plus que n augmente.
On reprend la même démarche avec une autre définition pour les xi :

```
> x2:=(i,n) -> 5*cos(Pi/(n+1)*(i+1/2));
```

$$x2 := (i, n) \to 5\cos\left(\frac{\pi\left(i+\frac{1}{2}\right)}{n+1}\right)$$

```
> plot([f(X),fLa(f,5,x2),fLa(f,15,x2),fLa(f,20,x2)],X=-5..5,0..1,color=[black,red,blu
  e,green],numpoints=200,thickness=3);
```



Le phénomène a disparu.
Reste à determiner le lien entre ces xi particuliers et les polynômes de Tchebycheff :

```
> with(orthopoly):
```

T(n,x) est le n-ième polynome de Tchebycheff en x.

```
> T(2,x);
```


$$2\,x^2-1$$

```
> T(8,x2(3,7)/5);
```

$$128\cos\left(\frac{7}{16}\pi\right)^8-256\cos\left(\frac{7}{16}\pi\right)^6+160\cos\left(\frac{7}{16}\pi\right)^4-32\cos\left(\frac{7}{16}\pi\right)^2+1$$

```
> evalf(%);
```

$$0$$

x2(i,n)/5 est racine de T(n+1).

Ex 3 : Racines rationnelles de polynômes de Z[X]

si p/q irréductible est racine de an*x^n+...+a0 alors :
an*p^n=-q*(a(n-1)*p^(n-1)+...+a0*q^n-1) c'est à dire :
q divise an*p^n. Or q est premier avec p et donc avec p^n et donc q divise an.
De même p divise a0.
Application : Faisons une fonction qui renvoie tous les nombres répondant à la condition nécessaire que l'on vient de démontrer.
Un exemple :

```
> poly:=27*(x-4/3)*(x-7/9);
```

$$poly := 27\left(x-\frac{4}{3}\right)\left(x-\frac{7}{9}\right)$$

```
> expand(poly);
```

$$27\,x^2-57\,x+28$$

```
> lcoeff(poly,x);
```

$$27$$

```
> tcoeff(poly,x);
```

$$28$$

```
> divisors(28);
```

$$\{1, 2, 4, 7, 14, 28\}$$

D'ou la fonction :

```
> possibles:= (p,x) ->
  {seq(seq(seq(s*i/j,j=divisors(lcoeff(p,x))),i=divisors(tcoeff(p,x))),s={-1,1})}:
Warning, `s` in call to `seq` is not local
Warning, `i` in call to `seq` is not local
Warning, `j` in call to `seq` is not local
> possibles(poly,x);
```

$$\left\{\frac{-7}{9}, \frac{-2}{3}, \frac{-7}{3}, \frac{-7}{27}, \frac{-14}{3}, \frac{-14}{9}, \frac{-14}{27}, \frac{-28}{3}, \frac{-28}{9}, \frac{-28}{27}, \frac{-4}{27}, \frac{1}{9}, \frac{1}{27}, \frac{2}{3}, \frac{2}{9}, \frac{2}{27}, \frac{4}{3}, \frac{4}{9}, \frac{4}{27}, \frac{1}{3}, -1, 1, -2, 2, -4, 4, -7, 7, -14, 14, -28, 28, \frac{14}{3}, \frac{14}{9}, \frac{14}{27}, \frac{28}{3}, \frac{28}{9}, \frac{7}{3}, \frac{7}{9}, \frac{7}{27}, \frac{-4}{3}, \frac{-1}{9}, \frac{-1}{27}, \frac{-4}{9}, \frac{28}{27}, \frac{-2}{9}, \frac{-2}{27}, \frac{-1}{3}\right\}$$

Reste à ne selectionner que les 'vraies' racines.
On crée une fonction qui renvoie true (vrai) si 'v' est une racine du polynôme 'p' d'indéterminée 'x' :

```
> estnul:=(v,p,x) -> subs(x=v,p)=0:
```

Et on utilise **select** pour ne prendre dans les possibles (poss) que les vraies racines :

```
> vraies:= (p,x,poss) -> select(estnul,poss,p,x):
```

On combine le tout :

```
> toutesracines := (p,x) -> vraies(p,x,possibles(p,x)):
```

Résultat :

```
> toutesracines(poly,x);
```

$$\left\{\frac{4}{3}, \frac{7}{9}\right\}$$